

Algebraic Manipulation for Formal Solutions

熊本大学 理学部 交野 雅代 (Masayo KATANO)

1. Introduction

The asymptotic expansion plays an important role in the numerical analysis of physical and engineering problems, in particular, in many problems treating differential equations. For example, in the calculation of approximate values of an entire solution for a sufficiently large variable, the asymptotic expansion is more effective than the Taylor expansion.

According to the well-established local theory of linear differential or difference equations, an actual solution near an irregular singular point is characterized by its asymptotic behavior. More precisely, one can find formal power series solutions, which are, in general, divergent, and then one can verify the existence of actual solutions admitting formal solutions as their asymptotic expansions.

Now, it is not difficult to find (understand theoretically) the existence of formal solutions, but to seek explicit forms of them needs a very tedious calculation. So, our objective is to construct an expert system of algebraic manipulation for the evaluation of formal solutions of linear differential equations. We follow the algorithms in the papers [1] and [2].

Consider the system of linear differential equations

$$t \frac{dX}{dt} = (A_0 + A_1 t + \cdots + A_q t^q) X \quad (1)$$

and the single linear differential equation

$$t^n \frac{dx}{dt^n} = \sum_{\ell=1}^n \left(\sum_{r=0}^{q\ell} a_{\ell,r} t^r \right) t^{n-\ell} \frac{d^{n+\ell}x}{dt^{n+\ell}}. \quad (2)$$

Those differential equations have an irregular singularity of Poicaré rank q at $t = \infty$. And they are of canonical and reduced forms of more general differential equations which have an irregular singularity.

Near the irregular singularity, one can find formal solutions of the form

$$X^k(t) = \exp \left(\frac{\lambda_k}{q} t^q + \frac{\alpha_{q-1}^k}{q-1} t^{q-1} + \cdots + \alpha_1^k t \right) t^{\mu_k} \sum_{s=0}^{\infty} H_k(s) t^{-s}. \quad (3)$$

The principal characteristic constants λ_k are given by eigenvalues of A_q in case (1), and by roots of the algebraic equation

$$J(\lambda) = \lambda^n - \sum_{\ell=1}^n a_{\ell,q\ell} \lambda^{n-\ell} = 0 \quad (4)$$

in case (2). Other characteristic constants α_j^k ($j = q-1, q-2, \dots, 1$) and μ_k are determined by algebraic processes.

Problems

1. How the α_j^k and μ_k are determined by the principal characteristic constant λ_k and the preceding characteristic constants $\alpha_{q-1}^k, \dots, \alpha_{j+1}^k$ in succession ?
2. How the coefficients $H_k(s)$ are calculated by the initial value $H_k(0)$?
3. How an identity called Fuchs relation between the characteristic exponents μ_k and those ρ_j at the origin :

$$\sum_{j=1}^n \rho_j = \sum_{k=1}^n \mu_k \quad (\text{in case (1)}),$$

$$\sum_{j=1}^n \rho_j = \sum_{k=1}^n \mu_k + q \binom{n}{2} \quad (\text{in case (2)}).$$

can be derived ?

We shall now construct an **expert system** of algebraic manipulation for such problems.

2. Algorithms for the first system of equations

We first consider the system of linear differential equations (1). The coefficient $H(s)$ satisfies the system of linear difference equations

$$(A_q - \lambda)H(s+q) + (A_{q-1} - \alpha_{q-1})H(s+q-1) + \dots + (A_0 - \mu + s)H(s) = 0 \quad (5)$$

subject to the initial condition that $H(r) = 0$ ($r < 0$).

In order to show that the coefficient $H(s)$ can be determined successively, we have only to reduce the system (5) of a *singular* type to that of a *regular* type, together with the determination of all characteristic constants.

By decomposing the vector $H(s)$ and the matrices A_k as follows:

$$H(s) = \begin{pmatrix} h(s) \\ \hat{H}(s) \end{pmatrix}, \quad A_k = \begin{pmatrix} a_k & \beta_k \\ \gamma_k & \hat{A}_k \end{pmatrix} \quad (k = 0, 1, \dots, q-1),$$

we rewrite the system (5) in the form

$$\sum_{k=1}^{q-1} (a_k - \alpha_k)h(s+k) + (a_0 - \mu + s)h(s) + \sum_{k=0}^{q-1} \beta_k \hat{H}(s+k) = 0 \quad (6)$$

and

$$\hat{H}(s+q) = \sum_{k=1}^{q-1} \mathcal{A}_k \hat{H}(s+k) + (\mathcal{A}_0 + \Lambda s) \hat{H}(s) + \sum_{k=0}^{q-1} \eta_k h(s+k), \quad (7)$$

where the β_k and the η_k denote $(n - 1)$ -dimensional row and column vectors, respectively, and the \mathcal{A}_k are $(n - 1)$ by $(n - 1)$ matrices. Moreover, Λ is a diagonal matrix

$$\Lambda = \text{diag}((\lambda - \lambda_2)^{-1}, (\lambda - \lambda_3)^{-1}, \dots, (\lambda - \lambda_n)^{-1})$$

and

$$\mathcal{A}_k = \Lambda(\hat{A}_k - \alpha_k), \quad \eta_k = \Lambda \gamma_k \quad (k = 0, 1, \dots, q - 1; \alpha_0 \equiv \mu).$$

Substituting (7) into (6) one after another, we calculate the underlined part:

$$\begin{aligned} \underline{\sum_{k=0}^{q-1} \beta_k \hat{H}(s+k)} &= \sum_{\ell=1}^{q-\nu} P(\nu : \ell) \hat{H}(s+q-\nu-\ell) + \sum_{\ell=1}^{\nu} Q(\nu : \ell : s) \hat{H}(s-\ell) \\ &\quad + \sum_{\ell=1}^{q-1} r_1(\nu : \ell) h(s-1+\ell) + \sum_{\ell=1}^{\nu} r_2(\nu : \ell) h(s-\ell) \\ &\quad (\nu = 1, 2, \dots, q), \end{aligned}$$

finally obtaining

$$\begin{aligned} \underline{\sum_{k=0}^{q-1} \beta_k \hat{H}(s+k)} &= \sum_{\ell=1}^{q-1} r_1(q : \ell) h(s-1+\ell) + \sum_{\ell=1}^q r_2(q : \ell) h(s-\ell) \\ &\quad + \sum_{\ell=1}^q Q(q : \ell : s) \hat{H}(s-\ell). \end{aligned}$$

Then, substituting this into (6), we consequently derive the required formulas of determining the characteristic constants and $h(s)$ as follows:

$$\begin{aligned} \sum_{k=1}^{q-1} (a_k - \alpha_k + r_1(q : k+1)) h(s+k) + (a_0 - \mu + r_1(q : 1) + s) h(s) \\ + \sum_{\ell=1}^q r_2(q : \ell) h(s-\ell) + \sum_{\ell=1}^q Q(q : \ell : s) \hat{H}(s-\ell) = 0 \quad (r_1(q : q) \equiv 0), \end{aligned}$$

whence

$$\begin{cases} \alpha_{q-1} = a_{q-1}, \\ \alpha_k = a_k + r_1(q : k+1) \quad (k = q-2, q-3, \dots, 1), \\ \mu = a_0 + r_1(q : 1), \end{cases} \quad (8)$$

and

$$s h(s) + \sum_{\ell=1}^q r_2(q : \ell) h(s-\ell) + \sum_{\ell=1}^q Q(q : \ell : s) \hat{H}(s-\ell) = 0. \quad (9)$$

From (7) and (9), we have thus obtained the regular system of linear difference equations for $H(s)$:

$$H(s+q) = B_{q-1}(s)H(s+q-1) + B_{q-2}(s)H(s+q-2) + \dots + B_0(s)H(s).$$

Algorithm of $Q(q : \ell : s)$ and $r_i(q : \ell)$ ($i = 1, 2$)

$$P(\nu + 1 : \ell) = P(\nu : 1)\mathcal{A}_{q-\ell} + P(\nu : \ell + 1) \quad (1 \leq \ell \leq q - \nu - 1), \quad (\text{I})$$

$$\begin{cases} Q(\nu + 1 : \ell : s) = P(\nu : 1)\mathcal{A}_{\nu+1-\ell} + Q(\nu : \ell : s) & (1 \leq \ell \leq \nu), \\ Q(\nu + 1 : \nu + 1 : s) = P(\nu : 1)(\mathcal{A}_0 + \Lambda(s - \nu - 1)), \end{cases} \quad (\text{II})$$

$$\begin{cases} r_1(\nu + 1 : \ell) = P(\nu : 1)\eta_{\nu+\ell} + r_1(\nu : \ell) & (1 \leq \ell \leq q - \nu - 1), \\ r_1(\nu + 1 : \ell) = r_1(\nu : \ell) & (q - \nu \leq \ell \leq q - 1), \end{cases} \quad (\text{III})$$

$$\begin{cases} r_2(\nu + 1 : \ell) = P(\nu : 1)\eta_{\nu+1-\ell} + r_2(\nu : \ell) & (1 \leq \ell \leq \nu), \\ r_2(\nu + 1 : \nu + 1) = P(\nu : 1)\eta_0, \end{cases} \quad (\text{IV})$$

subject to the initial conditions

$$P(0 : \ell) = \beta_{q-\ell} \quad (\ell = 1, 2, \dots, q) \quad \text{and} \quad r_1(0 : \ell) = 0 \quad (\ell = 1, 2, \dots, q - 1).$$

Algebraic Manipulation by REDUCE

%%% PROCEDURES %%%

```
%%%%%%%%%%%%%
PROCEDURE S1(K,I,J);
BEGIN
SCALAR PRO;
PRO:=1;
IF I=1 AND J=K THEN RETURN PRO$
IF J=1 AND I=K THEN RETURN PRO$
IF I NEQ 1 AND I NEQ K AND I=J THEN RETURN PRO$
PRO:=0;
RETURN PRO$
END;
%%%%%%%%%%%%%
OPERATOR LMD1,LMD2,LMD3; % LAM;
%%%%%%%%%%%%%
PROCEDURE LMD(K,I,J);
BEGIN
SCALAR PRO;
FOR L:=1:N DO
FOR M:=1:N DO
IF L NEQ K AND M NEQ K AND L=M THEN
LMD1(K,L,M):=1/(LAM(K)-LAM(L))
ELSE
LMD1(K,L,M):=0;
FOR L:=1:N DO
FOR M:=1:N DO
LMD2(K,L,M):=(FOR T:=1:N SUM
S1(K,L,T)*LMD1(K,T,M));
FOR L:=1:N DO
FOR M:=1:N DO
LMD3(K,L,M):=(FOR T:=1:N SUM
LMD2(K,L,T)*S1(K,T,M));
IF I<=N-1 AND J<=N-1 THEN RETURN
```

```

PRO:=LMD3(K,I+1,J+1)$
END;
%%%%%%%%%%%%%%%
OPERATOR A1; % SA;
%%%%%%%%%%%%%%%
PROCEDURE A(K,N,R,I,J);
BEGIN
SCALAR PRO;
FOR L:=1:N DO
FOR M:=1:N DO
A1(R,L,M):=(FOR T:=1:N SUM S1(K,L,T)*SA(R,T,M));
PRO:=(FOR T:=1:N SUM A1(R,I,T)*S1(K,T,J));
RETURN PRO$%
END;
%%%%%%%%%%%%%%%
PROCEDURE AH(K,N,R,I,J);
BEGIN
SCALAR PRO;
IF I<=N-1 AND J<=N-1 THEN RETURN
PRO:=A(K,N,R,I+1,J+1)$
END;
%%%%%%%%%%%%%%%
PROCEDURE BETA(K,N,R,J);
BEGIN
SCALAR PRO;
IF J<=N-1 THEN RETURN
PRO:=A(K,N,R,1,J+1)$
END;
%%%%%%%%%%%%%%%
PROCEDURE GAM(K,N,R,I);
BEGIN
SCALAR PRO;
IF I<=N-1 THEN RETURN
PRO:=A(K,N,R,I+1,1)$
END;
%%%%%%%%%%%%%%%
PROCEDURE ETA(K,N,R,I);
BEGIN
SCALAR PRO;
PRO:=(FOR T:=1:N-1 SUM LMD(K,I,T)*GAM(K,N,R,T));
RETURN PRO$%
END;
%%%%%%%%%%%%%%%
OPERATOR ALP;
%%%%%%%%%%%%%%%
PROCEDURE AA(K,N,R,I,J);
BEGIN
SCALAR PRO;
IF I=J THEN RETURN
PRO:=(FOR T:=1:N-1 SUM
LMD(K,I,T)*(AH(K,N,R,T,I)-ALP(K,R)))$%
PRO:=(FOR T:=1:N-1 SUM LMD(K,I,T)*AH(K,N,R,T,J));
RETURN PRO$%
END;

```

```

%%%%%%%%%%%%%
PROCEDURE P(K,N,Q,NU,R,J);
BEGIN
SCALAR PRO;
  IF NU=0 THEN RETURN
    PRO:=BETA(K,N,Q-R,J)$
  IF NU>=1 AND R<=Q-NU THEN RETURN
    PRO:=(FOR T:=1:N-1 SUM
      P(K,N,Q,NU-1,1,T)*AA(K,N,Q-R,T,J))
        +P(K,N,Q,NU-1,R+1,J)$
END;
%%%%%%%%%%%%%
PROCEDURE R1(K,N,Q,NU,R);
BEGIN
SCALAR PRO;
  IF NU=0 THEN RETURN
    PRO:=0$
  IF NU>=1 AND R<=Q-NU THEN RETURN
    PRO:=(FOR T:=1:N-1 SUM
      P(K,N,Q,NU-1,1,T)*ETA(K,N,NU-1+R,T))
        +R1(K,N,Q,NU-1,R)$
  IF NU>=1 AND R>=Q-NU+1 AND R<=Q-1 THEN RETURN
    PRO:=R1(K,N,Q,NU-1,R)$
END;
%%%%%%%%%%%%%
PROCEDURE R2(K,N,Q,NU,R);
BEGIN
SCALAR PRO;
  IF NU=0 THEN RETURN
    PRO:=0$
  IF NU>=1 AND R>=1 AND R<=NU-1 THEN RETURN
    PRO:=(FOR T:=1:N-1 SUM
      P(K,N,Q,NU-1,1,T)*ETA(K,N,NU-R,T))
        +R2(K,N,Q,NU-1,R)$
  IF NU>=1 AND R=NU THEN RETURN
    PRO:=(FOR T:=1:N-1 SUM
      P(K,N,Q,NU-1,1,T)*ETA(K,N,0,T))$
END;
%%%%%%%%%%%%%
OPERATOR S;
%%%%%%%%%%%%%
PROCEDURE QQ(K,N,Q,NU,R,S,J);
BEGIN
SCALAR PRO;
  IF NU=0 THEN RETURN
    PRO:=0$
  IF NU>=1 AND R>=1 AND R<=NU-1 THEN RETURN
    PRO:=(FOR T:=1:N-1 SUM
      P(K,N,Q,NU-1,1,T)*AA(K,N,NU-R,T,J))
        +QQ(K,N,Q,NU-1,R,S,J)$
  IF NU>=1 AND R=NU THEN RETURN
    PRO:=(FOR T:=1:N-1 SUM P(K,N,Q,NU-1,1,T)
      *(AA(K,N,0,T,J)+(S-NU)*LMD(K,T,J)))$
END;

```

```
%%%%%%%%%%%%%
PROCEDURE B(K,N,Q,R,S,I,J);
BEGIN
SCALAR PRO;
  IF R<0 OR R>=Q THEN RETURN
  PRO:=0$
  IF I=1 AND J=1 THEN RETURN
  PRO:=-(1/S)*R2(K,N,Q,Q,Q-R)$
  IF I=1 AND J>=2 THEN RETURN
  PRO:=-(1/S)*QQ(K,N,Q,Q,Q-R,S,J-1)$
  IF J=1 AND I>=2 THEN RETURN
  PRO:=ETA(K,N,R,I-1)$
  IF R=0 AND I>=2 AND J>=2 THEN RETURN
  PRO:=AA(K,N,0,I-1,J-1)+(S-Q)*LMD(K,I-1,J-1)$
  IF R>=1 AND R<=Q-1 AND I>=2 AND J>=2 THEN RETURN
  PRO:=AA(K,N,R,I-1,J-1)$
END;
%%%%%%%%%%%%%
PROCEDURE HH(K,N,Q,S,I);
BEGIN
SCALAR PRO;
  IF S<0 THEN RETURN
  PRO:=0$
  IF S=0 AND I=1 THEN RETURN
  PRO:=1$
  IF S=0 AND I>=2 THEN RETURN
  PRO:=0$
  IF S>=1 THEN RETURN
  PRO:=(FOR L:=1:Q SUM
        (FOR T:=1:N SUM
          B(K,N,Q,Q-L,S,I,T)*HH(K,N,Q,S-L,T)))$
END;
%%%%%%%%%%%%%
PROCEDURE H(K,N,Q,S,I);
BEGIN
SCALAR PRO;
  PRO:=(FOR T:=1:N SUM S1(K,I,T)*HH(K,N,Q,S,T));
RETURN PRO$
```

```
%%%%%%%%%%%%%
OPERATOR SA,LAM,MU,R1;
%%%%%%%%%%%%%
```

%%% MAIN PROGRAM %%%

```
FOR K:=1:N DO
  FOR R:=Q-1 STEP -1 UNTIL 0 DO
    BEGIN
      IF R=Q-1 THEN
        R1(K,N,Q,Q,R+1):=0;
```

```

ALP(K,R):=A(K,N,R,1,1)+R1(K,N,Q,Q,R+1);
END;
WRITE
" ***** SYSTEM OF LINEAR DIFFERENTIAL EQUATIONS *****";
WRITE " N = ",N," , Q = ",Q;
WRITE " *** ALP(K,R) ***";

FACTOR SA;
ON RAT;

FOR K:=1:N DO
BEGIN
  FOR R:=Q-1 STEP -1 UNTIL 1 DO WRITE
    " ALP(",K,",",R,") = ",ALP(K,R);
  WRITE " MU(",K,") = ",MU(K,0);
END;

OFF RAT;
REMFAC SA;

WRITE " *** FUCHS RELATION ***";
WRITE "      ",(FOR K:=1:N SUM MU(K)), " = "
      ,(FOR K:=1:N SUM ALP(K,0));
%%%%%%%%%%%%%
END;

```

Output

***** SYSTEM OF LINEAR DIFFERENTIAL EQUATIONS *****

N = 4 , Q = 2

*** ALP(K,R) ***

ALP(1,1) = SA(1,1,1)

$$MU(1) = \frac{- SA(1,4,1)*SA(1,1,4)}{LAM(4) - LAM(1)} +$$

$$\frac{- SA(1,3,1)*SA(1,1,3)}{LAM(3) - LAM(1)} +$$

$$\frac{- SA(1,2,1)*SA(1,1,2)}{LAM(2) - LAM(1)} + SA(0,1,1)$$

ALP(2,1) = SA(1,2,2)

$$MU(2) = \frac{- SA(1,4,2)*SA(1,2,4)}{LAM(4) - LAM(2)} +$$

$$\begin{aligned}
 & - \frac{\text{SA}(1,3,2)*\text{SA}(1,2,3)}{\text{LAM}(3) - \text{LAM}(2)} + \frac{\text{SA}(1,2,1)*\text{SA}(1,1,2)}{\text{LAM}(2) - \text{LAM}(1)} + \text{SA}(0,2,2) \\
 \text{ALP}(3,1) & = \text{SA}(1,3,3) \\
 \text{MU}(3) & = \frac{- \text{SA}(1,4,3)*\text{SA}(1,3,4)}{\text{LAM}(4) - \text{LAM}(3)} + \frac{\text{SA}(1,3,2)*\text{SA}(1,2,3)}{\text{LAM}(3) - \text{LAM}(2)} + \\
 & \frac{\text{SA}(1,3,1)*\text{SA}(1,1,3)}{\text{LAM}(3) - \text{LAM}(1)} + \text{SA}(0,3,3) \\
 \text{ALP}(4,1) & = \text{SA}(1,4,4) \\
 \text{MU}(4) & = \frac{\text{SA}(1,4,3)*\text{SA}(1,3,4)}{\text{LAM}(4) - \text{LAM}(3)} + \frac{\text{SA}(1,4,2)*\text{SA}(1,2,4)}{\text{LAM}(4) - \text{LAM}(2)} + \\
 & \frac{\text{SA}(1,4,1)*\text{SA}(1,1,4)}{\text{LAM}(4) - \text{LAM}(1)} + \text{SA}(0,4,4)
 \end{aligned}$$

*** FUCHS RELATION ***

$$\begin{aligned}
 \text{MU}(4) + \text{MU}(3) + \text{MU}(2) + \text{MU}(1) \\
 = \text{SA}(0,4,4) + \text{SA}(0,3,3) + \text{SA}(0,2,2) + \text{SA}(0,1,1)
 \end{aligned}$$

3. Algorithms for the second equation

We now consider the single linear differential equation (2), though expert systems of algebraic manipulations for the reduction of (2) to (1) are developed by S. Ohkohchi and M. Kohno.

In this case, it is not easy to derive a linear difference equation satisfied by the coefficient $H(s)$ only by the direct substitution of the formal solution (3) into (2). So we use the following algorithm: We put

$$x_p(t) = t^{-(q-1)p} \frac{d^p x}{dt^p} \equiv \exp(P(t)) t^\mu \sum_{s=0}^{\infty} h_p(s) t^{-s} \quad (p = 1, 2, \dots, n),$$

where

$$P(t) = \exp \left(\sum_{k=1}^q \frac{\alpha_k}{k} t^k \right) \quad (\alpha_q \equiv \lambda).$$

Then, from the relation

$$x_p(t) = t^{-(q-1)} x'_{p-1}(t) + (q-1)(p-1)t^{-q} x_{p-1}(t)$$

and the linear differential equation (2), we have

$$h_p(s) = \sum_{k=1}^q \alpha_k h_{p-1}(s - q + k) + (\mu - s + (q-1)p + 1)h_{p-1}(s - q), \quad (i)$$

$$h_n(s) = \sum_{\ell=1}^n \sum_{r=0}^{q\ell} a_{\ell,r} h_{n-\ell}(s + r - q\ell), \quad (ii)$$

where

$$h_0(s) \equiv H(s).$$

First, from (i) we attempt to express the $h_p(s)$ in the form

$$h_p(s) = \sum_{\nu=0}^{qp} M(p : \nu : s) H(s - \nu) \quad (p = 1, 2, \dots, n). \quad (10)$$

Then, substituting these formulas into (ii), we obtain the required linear difference equation for $H(s)$ as follows:

$$\sum_{\nu=0}^{nq} \left\{ M(n : \nu : s) - \sum_{r=0}^{nq} \sum_{\ell=1}^n a_{\ell,q\ell-r} M(n - \ell : \nu - r : s - r) \right\} H(s - \nu) = 0.$$

Here the first q coefficients are vanishing and such relations determine the characteristic constants $\lambda, \alpha_{q-1}, \dots, \alpha_1$:

$$\begin{aligned} M(n : \nu : s) - \sum_{r=0}^{nq} \sum_{\ell=1}^n a_{\ell,q\ell-r} M(n - \ell : \nu - r : s - r) &= 0 \\ \implies \alpha_{q-\nu} \quad (0 \leq \nu \leq q-1) \end{aligned}$$

and moreover,

$$\begin{aligned} M(n : q : q) - \sum_{r=0}^{nq} \sum_{\ell=1}^n a_{\ell,q\ell-r} M(n - \ell : q - r : q - r) &= 0 \\ \implies \mu. \end{aligned}$$

After that, we finally obtain the $q(n-1)$ -th order linear difference equation of a regular type for $H(s)$:

$$(s - q)J'(\lambda)H(s - q) = \sum_{\nu=q+1}^{nq} \{M(n : \nu : s) - \dots\} H(s - \nu).$$

In order to carry out the above calculation, we have only to seek the coefficients $M(p : \nu : s)$ in (10) by means of the following algorithm.

Algorithm of $M(p : \nu : s)$

$$M(p : \nu : s) = \sum_{k=0}^{\nu} \alpha_{q-k} M(p-1 : \nu-k : s-k)$$

$$(0 \leq \nu \leq q-1; \alpha_q \equiv \lambda),$$

$$\begin{aligned} M(p : \nu : s) &= \sum_{k=0}^{q-1} \alpha_{q-k} M(p-1 : \nu-k : s-k) \\ &\quad + (\mu - s + (q-1)p + 1) M(p-1 : \nu-q : s-q) \\ &\quad (q \leq \nu \leq q(p-1)), \\ M(p : \nu : s) &= \sum_{k=\nu-q(p-1)}^{q-1} \alpha_{q-k} M(p-1 : \nu-k : s-k) \\ &\quad + (\mu - s + (q-1)p + 1) M(p-1 : \nu-q : s-q) \\ &\quad (q(p-1) + 1 \leq \nu \leq qp). \end{aligned}$$

```
%%%%%%%%%%%%%
OPERATOR MU,S,ALP,X,J,A,LAM,TH;
%%%%%%%%%%%%%
PROCEDURE M(K,Q,P,NU,S);
BEGIN
SCALAR PRO;
ALP(K,Q):=X;
IF NU<0 OR NU>Q*P THEN RETURN
PRO:=0$;
IF NU=0 AND P=0 THEN RETURN
PRO:=1$;
IF NU>=0 AND NU<=Q-1 THEN RETURN
PRO:=(FOR T:=0:NU SUM
ALP(K,Q-T)*M(K,Q,P-1,NU-T,S-T))$;
IF NU>=Q AND NU<=Q*(P-1) THEN RETURN
PRO:=(FOR T:=0:Q-1 SUM
ALP(K,Q-T)*M(K,Q,P-1,NU-T,S-T))
+(MU(K)-S+(Q-1)*P+1)*M(K,Q,P-1,NU-Q,S-Q)$;
IF NU>=Q*(P-1)+1 AND NU<=Q*P THEN RETURN
PRO:=(FOR T:=NU:Q*P-1 SUM ALP(K,Q*P-T)
*M(K,Q,P-1,NU+Q*(P-1)-T,S+Q*(P-1)-T))
+(MU(K)-S+(Q-1)*P+1)*M(K,Q,P-1,NU-Q,S-Q)$;
END;
%%%%%%%%%%%%%
```

4. Examples

4.1. An important proposition

As an application of our system of algebraic manipulation, we can easily verify and calculate the explicit values in the following proposition, which played an important role in obtaining the estimates of $H(s)$ for sufficiently large values of s in [2]:

Proposition *Let s tend to the positive infinity. Then*

$$\lim_{s \rightarrow \infty} \frac{1}{s^k} M(p : kq : s) = (-1)^k \lambda^{p-k} \binom{p}{k} \quad (0 \leq k \leq p)$$

and

$$\lim_{s \rightarrow \infty} \frac{1}{s^k} M(p : kq + \ell : s) = c_{p,kq+\ell}$$

$$(0 \leq k \leq p-1, 1 \leq \ell \leq q-1).$$

%%%% MAIN PROGRAM %%%%

```

OPERATOR CC;
%%%%%%%%%%%%%
PROCEDURE TH1(K,Q,P,R,S);
BEGIN
SCALAR PRO;
  U:=REMAINDER(R,Q);
  IF U=0 THEN J:=R/Q
  ELSE      J:=(R-U)/Q;
  CC(K,R):=COEFFN(M(K,Q,P,R,S),S,J);
  LET X=LAM(K);
  PRO:=CC(K,R);
  CLEAR X;
IF R>=0 AND R<=P*Q THEN RETURN PRO$
END;
%%%%%%%%%%%%%

WRITE
  " ***** PROOF OF PROPOSITION *****";
WRITE " p = ",N,",", q = ",Q;
FOR I:=0:N DO
  BEGIN
    R:=I*Q;
    IF I=0 THEN
      WRITE " C_{",N,",",0} = ",TH1(K,Q,N,R,S)
    ELSE
      WRITE " C_{",N,",",",",I,"*",Q,"} = ",TH1(K,Q,N,R,S);
    END;
  WRITE " ";
FOR J:=1:Q-1 DO
  FOR I:=0:N-1 DO
    BEGIN
      R:=I*Q+J;
      IF I=0 THEN
        WRITE " C_{",N,",",",",J,"} = ",TH1(K,Q,N,R,S)
      ELSE
        WRITE " C_{",N,",",",",I,"*",Q,"+",J,"} = "
              ,TH1(K,Q,N,R,S);
    END;
  WRITE " ";
END;
%%%%%%%%%%%%%
***** PROOF OF PROPOSITION *****
p = 5 , q = 3
C_{5,0} = LAM(1)5
```

$$C_{\{5,1*3\}} = - 5*LAM(1)^4$$

$$C_{\{5,2*3\}} = 10*LAM(1)^3$$

$$C_{\{5,3*3\}} = - 10*LAM(1)^2$$

$$C_{\{5,4*3\}} = 5*LAM(1)$$

$$C_{\{5,5*3\}} = -1$$

$$C_{\{5,1\}} = 5*ALP(1,2)*LAM(1)^4$$

$$C_{\{5,1*3+1\}} = - 20*ALP(1,2)*LAM(1)^3$$

$$C_{\{5,2*3+1\}} = 30*ALP(1,2)*LAM(1)^2$$

$$C_{\{5,3*3+1\}} = - 20*ALP(1,2)*LAM(1)$$

$$C_{\{5,4*3+1\}} = 5*ALP(1,2)$$

$$C_{\{5,2\}} = 5*LAM(1)^3 * (2*ALP(1,2)^2 + ALP(1,1)*LAM(1))$$

$$C_{\{5,1*3+2\}} = - 10*LAM(1)^2 * (3*ALP(1,2)^2 + 2*ALP(1,1)*LAM(1))$$

$$C_{\{5,2*3+2\}} = 30*LAM(1)*(ALP(1,2)^2 + ALP(1,1)*LAM(1))$$

$$C_{\{5,3*3+2\}} = - 10*(ALP(1,2)^2 + 2*ALP(1,1)*LAM(1))$$

$$C_{\{5,4*3+2\}} = 5*ALP(1,1)$$

4.2. The Airy equation

As an example of illustrating our system of algebraic manipulation, we here deal with the Airy equation

$$\frac{d^2y}{dz^2} - zy = 0,$$

which is transformed to

$$t^2 y'' + \frac{1}{3} t y' - t^2 y = 0 \quad (11)$$

by the change of variables

$$t = \frac{2}{3} z^{\frac{3}{2}}.$$

Moreover, by putting $Y = (y_1, y_2)^T$, where

$$\begin{cases} y_1 = t y, \\ y_2 = t y', \end{cases}$$

we can rewrite (11) in the Birkhoff system

$$t Y' = \left\{ \begin{pmatrix} 1 & 0 \\ 0 & 2/3 \end{pmatrix} + t \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix} \right\} Y. \quad (12)$$

As for the system (12), using the first program, we have

$$\begin{aligned} \text{LAM}(1) &:= 1 \quad \text{MU}(1) = \frac{5}{6} \quad H(0) = 1 \quad H(1) = \frac{5}{72} \quad H(2) = \frac{385}{10368} \\ H(3) &= \frac{85085}{2239488} \quad H(4) = \frac{37182145}{644972544} \quad H(5) = \frac{5391411025}{46438023168} \\ \text{LAM}(2) &:= -1 \quad \text{MU}(1) = \frac{5}{6} \quad H(0) = 1 \quad H(1) = -\frac{5}{72} \quad H(2) = \frac{385}{10368} \\ H(3) &= -\frac{85085}{2239488} \quad H(4) = -\frac{37182145}{644972544} \quad H(5) = -\frac{5391411025}{46438023168} \end{aligned}$$

As for the single equation (11), it is easy see that the coefficients $h(s)$ satisfy

$$(\lambda^2 - 1)h(s+2) + 2\lambda(-s-1 + \mu + \frac{1}{6})h(s+1) + (s-\mu)(s-\mu + \frac{2}{3})h(s) = 0$$

and are therefore given by

$$h(s) = \left(\frac{1}{2\lambda}\right)^s \frac{\Gamma(s-\mu)\Gamma(s-\mu + \frac{2}{3})}{\Gamma(s+1)\Gamma(-\mu)\Gamma(-\mu + \frac{2}{3})}.$$

Combining PROCEDURE M(K,Q,P,NU,S) with other procedures, which we have not stated because of the limitation of pages, we have

$$\begin{aligned} \text{COEFF } H(S+2) &= \text{LAM}(1)^2 - 1 \\ \text{COEFF } H(S+1) &= \frac{\text{LAM}(1)*(6*\text{MU}(1) - 6*S - 5)}{3} \\ \text{COEFF } H(S) &= \frac{(3*(\text{MU}(1) - S - 1) + 1)*(\text{MU}(1) - S)}{3} \end{aligned}$$

References

- [1] M. Kohno: A two point connection problem for general linear ordinary differential equations, *Hiroshima Math. J.*, 4 (1974), 293–338.
- [2] M. Kohno: A two point connection problem, *Hiroshima Math. J.*, 9 (1979), 61–135.